TP Maple 2 | Nombres

1. Les nombres

Dans ce chapitre, nous exposerons la gestion des nombres (entiers, rationnels, réels, complexes) par Maple. Les commandes les plus importantes seront consignées dans des tableaux récapitulatifs.

1.1. Les entiers et l'arithmétique

Maple peut manipuler des nombres comportant un très grand nombre de chiffres mais néanmoins limité. Le nombre maximal de chiffres, *maxdigits*, dépend du matériel utilisé. On y aura accès par la commande **kernelopts(maxdigits)**.

```
> kernelopts(maxdigits);
```

$$268435448$$

Voici un aperçu des différentes opérations possibles sur les nombres entiers (« integers » en anglais) :

► *La fonction factorielle* : pour tout $n \in \mathbb{N}$, $n! = \prod_{k=1}^{n} k$ se calcule au moyen de la commande **factorial** ou à l'aide du symbole **!**.

```
> factorial(12);
```

$$479001600$$

```
> 12!;
```

$$479001600$$

▶ *Division euclidienne* : on utilise les commandes **irem** et **iquo** pour calculer le reste et le quotient de la division euclidienne (**irem** et **iquo** sont les abréviations respectives de *integer remainder* et *integer quotient*).

```
> irem(23,3);
```

$$2$$

```
> iquo(23,3);
```

$$7$$

▶ *Test de primalité* : la fonction **isprime** permet de tester si un entier est premier ou non.

```
> isprime(23);
```

$$\mathit{true}$$

▶ *Décomposition en produit de facteurs premiers* : Il s'agit du théorème fondamental de l'Arithmétique, à savoir tout nombre naturel plus grand que 1 se décompose en produit unique de facteurs premiers. La fonction **ifactor** calcule la décomposition en produit de facteurs premiers d'un entier donné.

```
> ifactor(561);
```

$$(3)(11)(17)$$

On a donc $561 = 3 \times 11 \times 17$.

Arithmétique élémentaire sous Maple

- **irem(n,m)** : renvoie le reste de n dans la division euclidienne par m.
- **iquo(n,m)** : renvoie le quotient de n dans la division euclidienne par m.
- **igcd(n,m)** : renvoie pgcd(n, m).
- **ilcm(n,m)** : renvoie ppcm(n, m).
- **ithprime(n)** : renvoie le n-ième nombre premier.
- **isprime(n)** : renvoie *true* si n est premier, *false* sinon.
- **ifactor(n)** : renvoie la décomposition en produit de facteur(s) premier(s) de n.

Attention, lancez **ithprime**(10^6), puis **ithprime**(10^7) ou **ithprime**(10^8). Vous avez intérêt de savoir où se trouve le bouton « Interrupt the current operation ».

1.2. Les nombres rationnels

Maple manipule les nombres rationnels formellement, c'est-à-dire comme des quotients d'entiers et non pas des approximations décimales. En cas de calcul sur des nombres rationnels, le logiciel renverra une écrire du résultat sous forme irréductible, ie sous la forme d'une fraction *simplifiée*.

```
> (1/2-1/3)*(66/78);
```

$$\frac{11}{78}$$

1.3. Nombres remarquables

Au-delà des nombres rationnels, Maple sait manipuler formellement certains nombres irrationnels.

Les radicaux

C'est le cas des racines carrées de nombres rationnels. La racine carrée d'un nombre r s'obtient par la commande [1] **sqrt(r)**.

```
> x:=sqrt(2):x^2;
```

$$2$$

Le nombre irrationnel [2] $\sqrt{2}$ est traité par le logiciel comme l'unique solution positive de l'équation $x^2 = 2$ et non pas comme une approximation décimale. Il faudra parfois forcer la main au logiciel pour obtenir certaines simplifications pour nous évidentes.

1. **sqrt** comme abréviation de *square root*, *racine carrée* in French.
2. Voir le cours de Mathématique pour la preuve (classique) de l'irrationnalité de $\sqrt{2}$.

```
> sqrt(10)/sqrt(2);
```

$$\sqrt{(10)}\sqrt{(2)}$$

```
> simplify(%);
```

$$\sqrt{5}$$

La commande **simplify** permet de telles simplifications lors de calculs symboliques. On pourra également utiliser les commandes **radnormal** et **rationalize** permettant des simplifications.

```
> a:=(7+5*sqrt(2))^(1/3): radnormal(a);
```

$$1+\sqrt{2}$$

```
> rationalize(1/(sqrt(2)+sqrt(3)+sqrt(5));
```

$$-\left(\frac{1}{12}\left(-\sqrt{2}-\sqrt{3}+\sqrt{5}\right)\right)\sqrt{2}\sqrt{3}$$

Simplifications des radicaux

▶ **simplify** : permet de symplifier des expressions radicalaires (entre autres) ; pour les expressions rebelles e, on pourra utiliser **simplify(e,radical)** ou **simplify(e,sqrt)** (plus efficace s'il n'y a que des racines carrées).

▶ **rationalize** : simplifie des expressions où apparaissent des radicaux au dénominateur.

▶ **radnormal** : simplifie des expressions où apparaissent des radicaux ; pour éviter des radicaux au dénominateur, on l'utilisera avec l'option 'rationalized'.

1.4. Les nombres décimaux

On appelle nombre décimal tout nombre rationnel qui est le quotient d'un entier relatif par une puissance entière de 10. Les décimaux sont notés sous Maple à l'anglo-saxonne : avec un point **.** au lieu de la virgule **,** française. Ces nombres sont qualifiés de *flottants* [3] et leur type est *float*.

```
> 1.3-1.5;
```

$$-0.2$$

Il y a trois écritures possibles d'un décimal au moyen des puissances de 10 :

▶ *la notation dite scientifique* : au moyen du symbole e ou E (indifféremment) ;

3. C'est en fait la virgule qui est flottante !

- *les puissances usuelles de 10 ;*
- *la fonction* **Float**.

Par exemple, le décimal 0.0012 pourra être défini par

$1.2e-3$, $1.2E-3$, $1.2*10^{-3}$ ou encore **Float(12,-4)**. [4]

On retiendra la règle suivante :

Calcul formel vs. évaluation décimale

- Le mélange dans une expression de nombres sous forme de fraction et sous forme décimale entraîne l'écriture du résultat sous forme décimale ;
- Lorsqu'une fonction est appelée avec un argument sous forme décimale, le résultat retourné sera sous forme décimale ;
- En revanche, priroité est donnée au calcul symbolique sur le calcul numérique.

De quoi éclairer ce qui suit :

```
> 1/2+0.25;
```

$$0.75$$

```
> 1/2+1/4;
```

$$\frac{3}{4}$$

```
> sqrt(0.25);
```

$$0.5000000000$$

```
> sqrt(1/4);
```

$$\frac{1}{2}$$

```
> 1.024+Pi;
```

$$1.024+\pi$$

Signalons que l'on peut convertir un nombre décimal d donné en fraction irréductible en utilisant la commande **convert(d,rational)**.

4. Les arguments de cette fonction doivent être des entiers sous peine d'erreurs.

```
> convert(3.1415927,rational);
```

$$\frac{126003}{40108}$$

Ecriture fractionnaire d'un nombre décimal

convert(d,rational) : donne la forme fractionnaire irréductible du nombre décimal d.

1.5. Les nombres réels

Nous avons déjà remarqué que Maple effectue *en priorité* des calculs exacts sur les fractions. Il faudra donc utiliser une commande spécifique afin d'obtenir des valeurs approchées sous forme décimale.

```
> sqrt(2);
```

$$\sqrt{2}$$

C'est la fonction **evalf** qui permet d'obtenir des valeurs approchées sous forme décimale [5] avec un nombre de chiffres significatifs contenu dans la variable *globale* **Digits**. La valeur par défaut de **Digits** est 10. Attention, si l'on modifie cette variable, la nouvelle valeur sera utilisée pour toutes les évaluations qui suivront dans la feuille de calcul.

```
> evalf(sqrt(2));
```

$$1.414213562$$

```
> Digits:=20:evalf(sqrt(2));
```

$$1.4142135623730950488$$

On peut obtenir le même résultat en spécifiant le nombre de chiffres significatifs voulu en second argument de **evalf**. Cette méthode a l'avantage de ne changer que *localement* la valeur de **Digits**.

```
> evalf(sqrt(2),20);
```

$$1.4142135623730950488$$

Rappelons pour finir que la présence d'un nombre sous forme décimale en argument d'une fonction forcera Maple à écrire le résultat sous forme décimale [6].

5. Le nom **evalf** est une contraction de *floating evaluation*.
6. Avec toutefois priorité au calcul symbolique.

```
> sqrt(2.1);
```

$$1.449137674$$

Manipulation des flottants

- **Digits := n** : fixe à n le nombre de chiffres significatifs lors de l'appel de **evalf**.
- **evalf(x)** : renvoie une valeur décimale approchée de x avec **Digits** chiffres significatifs.
- **evalf(x,n)** : renvoie une valeur décimale approchée de x avec n chiffres significatifs.
- **floor(x)** : partie entière de x.
- **round(x)** : entier le plus proche de x.

1.6. Les nombres complexes

Sous Maple, tout nombre complexe z s'écrit sous forme algébrique $z = a + b * I$ où le symbole **I** désigne le nombre noté i dans le cours de Mathématiques. Ce symbole est *réservé* : au même titre que **Pi**, on ne peut pas changer sa signification sous peine d'erreur. Maple effectue les opérations algébriques usuelles sur les nombres complexes. La multiplication sera codée explicitement par le signe *.

```
> (3 + I)*(2*I + 3);
```

$$7 + 9I$$

Dans le cas où Maple ne retourne pas une écriture sous forme algébrique, on pourra lui forcer la main en utilisant la fonction **evalc**.

```
> (sqrt(2) + I)*(1 + I);
```

$$(\sqrt{2} + I)(1 + I)$$

```
> evalc((sqrt(2) + I)*(1 + I));
```

$$\sqrt{2} - 1 + I(1 + \sqrt{2})$$

On peut également définir un nombre complexe sous forme trigonométrique à l'aide de la fonction **polar**. Le passage d'une forme algébrique à une forme trigonométrique s'effectue au moyen des commandes **argument** et **abs** calculant respectivement un argument appartenant à $]-\pi, \pi]$ et le module d'un nombre complexe.

```
> evalc(polar(1,Pi/3));
```

$$\frac{1}{2} + \frac{1}{2} I\sqrt{3}$$

Manipulation des nombres complexes

- **z^n** : renvoie z^n.
- $\mathbf{z_1} * \mathbf{z_2}$: renvoie $z_1 \times z_2$.
- **abs(z)** : renvoie $|z|$.
- **argument(z)** : renvoie un argument de z. Attention, sous Maple, 0 est d'argument 0 !
- **Re(z)** : renvoie $\mathrm{Re}(z)$.
- **Im(z)** : renvoie $\mathrm{Im}(z)$.
- **polar(r,θ)** : renvoie la valeur $re^{i\theta}$.
- **conjugate(z)** : renvoie $\overline{z}$.

2. La nature des nombres

Il est utile de connaître le type des nombres que l'on manipule. Cela peut se faire par la commande **whattype**. Posez les questions suivantes.

```
> whattype(10); whattype(10E1);whattype(10/4);whattype(Pi);whattype(exp(1));whattype(2-I);
```

3. La commande assume

Il est parfois important de faire des hypothèses sur les inconnues que l'on manipule. Commençons par un exemple.

```
> restart: z:=x+I*y: Re(z);
```

$$\mathrm{Re}(x + Iy)$$

Le logiciel ne sait pas calculer $\mathrm{Re}(z)$ pour la même raison que nous : les nombres x et y sont-ils réels ? Pour indiquer à Maple que x et y sont réels, on utilisera la commande **assume**.

```
> restart: assume(x,real,y,real): z:=x+I*y: Re(z);
```

$$x\text{~}$$

On remarquera que le logiciel affuble x d'un tilde ~ afin que l'on se souvienne que des hypothèses ont été faites sur x. On retiendra la syntaxe suivante.

La commande assume

assume(x, type) ou **assume(hypothèses sur** x **)**.

```
> restart: expand(ln(x*y));
```

$$\ell\text{n}(xy)$$

```
> assume(x>0,y>0): expand(ln(x*y));
```

$$\ell\text{n}(x\text{~}) + \ell\text{n}(y\text{~})$$

4. Exercices

Exercice 1.

Pour tout entier naturel n, on pose

$$u_n = \frac{1}{\binom{2n}{n}}.$$

1. Simplifier, pour tout entier naturel n,

$$v_n = \frac{u_{n+1}}{u_n}.$$

2. En déduire la limite de v_n quand n tend vers $+\infty$.

Exercice 2.

Simplifier

$$x = \sqrt[5]{13452\sqrt{2} - 19024}.$$

Exercice 3.

Combien y-a-t-il de zéros à la fin de l'écriture décimale de 300! ? On pourra utiliser la commande **ifactor**.

Exercice 4.

Simplifier les expressions suivantes :

$$A = \frac{1 + i\tan(\alpha)}{1 - i\tan(\alpha)} \quad \text{et} \quad B = \frac{(1+i)^n}{(1-i)^{n-2}}.$$

où $\alpha \in \mathbb{R}$ et $n \in \mathbb{Z}$.

Exercice 5.

On souhaite retrouver au moyen de Maple la célèbre formule de Machin :

$$\frac{\pi}{4} = 4\arctan\left(\frac{1}{5}\right) - \arctan\left(\frac{1}{239}\right).$$

1. On pose

$$z = (1-i)(5+i)^4.$$

1.a. Ecrire z sous forme algébrique.

1.b. Calculer un argument de z au moyen de la fonction $\arctan$.

1.c. En déduire que

$$\frac{\pi}{4} = 4\arctan\left(\frac{1}{5}\right) - \arctan\left(\frac{1}{239}\right) \quad [2\pi]$$

2. Conclure en utilisant **evalf**.

Exercice 6.

En vous aidant de Maple, trouver une simplification de la somme :

$$\arctan(2) + \arctan(3) + \arctan(2+\sqrt{3}).$$

Exercice 7.

Denote by $M_n = 2^n - 1$ the nth *Mersenne number*.

1. Prove that if M_n is a prime number then n is prime too. (Use a geometric series for this question.)
2. Is the converse true ? (Use Maple for this question.)

Exercice 8.

1. Grâce à Maple démontrer les formules

$$\sum_{k=1}^{n} k = \frac{n(n+1)}{2}, \qquad \sum_{k=1}^{n} k^2 = \frac{n(n+1)(2n+1)}{6}, \quad n \in \mathbb{N}.$$

2. Soit f_p la fonction définie par $f_p(n) = \sum_{k=1}^{n} k^p$ avec $r \in \mathbb{N}$. De quel type de fonction s'agit-il ? (Faire une conjecture à l'aide de Maple.)
3. Prouvez votre conjecture.

Solutions des exercices

Solution 1.

```
> u:=n->1/binomial(2*n,n);     v:=n->u(n+1)/u(n);
  simplify(expand((v(n))));    limit(v(n),n=infinity);
```

Solution 2.

```
> radnormal((13452*sqrt(2)-19024)^(1/5));
```

Solution 3.

Il y a 74 zéros. Il suffit de voir les exposants de 2 et de 5 dans la décomposition en facteurs premiers.

```
> ifactor(300!);
```

Solution 4.

```
> A:=simplify(evalc((1+I*tan(x))/(1-I*tan(x))));
  B:=expand(((1+I)/(1-I))^n*(1-I)^2);
```

Solution 5.

```
> z:=(1-I)*(5+I)^4: expand(z);
```

$$956 + 4I$$

```
> (4/956);  evalf((4*arctan(1/5)-arctan(4/956)));
```

$$\frac{1}{239}\,, \qquad 0.7853981635$$

Comme 0.7853981635 appartient à l'intervalle $[0, 2\pi[$ l'équivalence de l'énoncé est en fait une égalité.

Solution 6.

```
> y:=arctan(2) + arctan(3) + arctan(2 + sqrt(3)):  simplify(y);
```

Solution 7.

1. We prove the contraposition. Let $n = km$ be a natural number which is not prime, $1 < k < n$. Then

$$1 + 2^k + \left(2^k\right)^2 + \cdots + \left(2^k\right)^{m-1} = \frac{2^{km} - 1}{2^k - 1}\,.$$

This shows that M_n can be divided by $2^k - 1$, a number which is bigger than 1 and smaller than M_n ; thus M_n is not prime.

2. `> ifactor(2^11-1);` gives $M_{11} = 2047 = 23 \times 89$. So the converse is not true.

Solution 8.